
DLG Conversion to the SDSFIE using ESRI 

Overview

The User Guide presents a methodology for translating United States Geological Survey (USGS) digital line graph (DLG) files into Environmental Systems Research Institute (ESRI) ARC/INFO geographic information system coverages.  The coverages created with the source data will be compliant with the CADD/GIS Technology Center Spatial Data Standards (SDS/FMS). This document guides the user step by step through the procedures to perform bulk loading of source files to create a collection of coverages.  These tools are developed with Arc Macro Language (AML) to allow the user of ARC/INFO Workstation version 7.x and 8.0.x to utilize the code and further customize the tools as desired. The following section is the step by step process to accomplish this Step.

Digital Line Graph to ESRI (ARCINFO) Workflow Flowchart

This work flow chart details the steps necessary for the successful migration of DLG data into the MGE SDS/FMS data format.  All tasks must be successfully completed in order to assure data integrity. 

 Figure 1 Digital Line Graph Workflow Flowchart
Step 1
Create Directory Structure for DLG2SDS

Create a directory named dlg2sds and save the uncompressed DLG source files to this directory. The source files will have a *.cdo or *.opt file extension.

Step 2
Place Required Files in DLG2SDS Directory

The following collection of files is required for the conversion process:

*.AML


12 Arc Macro Language script files for the conversion process.


Detailed description of each aml file is provided in Appendix A.


batch.aml



cdo2dlgren.aml



d2sarc.aml



d2spoint.aml



d2spoly.aml



d2sds.aml



dlg2sdscon.aml



dlgsepmmcodes.aml



dlgsnc.aml



e00exporter.aml



opt2dlgren.aml



remove_blanks.aml

*.MENU


dlggui.menu - Menu file for the graphic user interface

*.SDS


polylist.sds - List of all SDS/FMS compliant names for polygon coverages

linelist.sds - List of all SDS/FMS compliant names for line coverages

pointlist.sds - List of all SDS/FMS compliant names for point coverages

*.DBF

The database format files named sds#.dbf serve as the mechanism to relate the major and minor codes found in the DLG source files to SDS/FMS compliant entity class names and coverage names. The field value used to join the SDS/FMS tables with the appropriate coverage INFO attributes tables (*.PAT and *.AAT) is made of up major minor value pairs. Simply a combination of the text strings found in the major# and minor#, field, separated by the dash character ‘-‘ for each individual feature record in the data set. Attribution is made from major1-minor1 through major5-minor5.

sds1.dbf - lookup table of major1–minor1 value pairs for SDS/FMS entity name and coverage name

sds2.dbf - lookup table of major2–minor2 value pairs

sds3.dbf - lookup table of major3–minor3 value pairs

sds4.dbf - lookup table of major4–minor4 value pairs

sds5.dbf - lookup table of major5–minor5 value pairs

Place all of the required conversion files and DLG source files into the dlg2sds directory. An example of the completed directory structure is shown in Figure 2.

Figure 2 Example of a completed installation directory structure

Step 3
Start ARC/INFO

Start ARC/INFO Workstation software and navigate to the DLG2SDS directory. At the Arc: prompt, enter workspace <drive>:/dlg2sds.  In the example shown in Figure 3, the data are stored and the AML will be running from the N:/ drive

Figure 3 ARC Command Prompt Window
At the Arc: prompt, enter &run dlg2sdscon. This command will open a pull down menu that will be used to perform the remainder of the conversion process. A diagram of the pull down menu is shown in Figure 4.

Figure 4 DLG2SDSCON Pulldown Menu
Step 4
Rename DLG Source files 

It is necessary to rename the source files from their original names (having a *.cdo or *.opt) to the extension which ARC/INFO is expecting for files in DLG format (*.dlg). The process of renaming all the files in the DLG2SDS directory is accomplished by using the Rename item option from the pulldown menu. Select Renameà*.cdo to *.dlg to rename the *.cdo files. Select Renameà*.opt to *.dlg to rename the *.opt files. If you have a collection of *.cdo and *.opt files, you must select both menu options to prepare all of the files for conversion. These menu options for the rename procedure are shown in Figure 5.

Figure 5 Rename Source Files
Step 5
Convert DLG Source files 

The main conversion batch process to create SDS/FMS compliant coverages from the DLG source data is started by using the Conversion item from the pulldown menu.  Selecting ConvertàDLG to SDS will begin the execution of six AML scripts to perform the following functions:

1.
Convert the graphic spatial data in DLG format to ARC/INFO point, line, and polygon coverages.

2.
Compile the major#-minor# attribute codes and identify relationships to the SDS/FMS entity types.

3.
Perform data set naming as appropriate to the features found in the source files and compliance with SDS/FMS.

4. Create ARC/INFO export files (*.e00) (exported with the –none option) of the new SDS/FMS coverages for distribution.

Figure 6 Convert DLG to SDS/FMS

Step 6
Delete original DLG Source files (optional)

If desired, the user can delete the DLG format source files that are used in the process. These files are not of immediate use to the ARC/INFO software and can be quickly removed to conserve disk space. For archive purposes it is advised that a user retain the original compressed version of the DLG data sets if another conversion of the same information is desired. To delete all *.dlg files in the DLG2SDS directory, select DeleteàSource DGL files from the pulldown menu.

Figure 7 Delete Original Source Files
Step 7
Quit DLG2SDS Conversion Pulldown Menu

When the conversion process is complete, you can dismiss the pulldown menu by clicking the Quit item. The user will return to the Arc: prompt in ARC/INFO. You can confirm the process is complete when you see the following message in the ARC/INFO command prompt window. 

********* DLG TO SDS CONVERSION PROCESS COMPLETE *********

No other interaction with the AMLs is required.

After dismissing the pulldown menu, it is suggested that the user move the completed coverages to another drive or directory on the GIS workstation. If an attempt is made to convert other data sets in the dlg2sds directory before removing the most recently created coverages, ARC/INFO will attempt to write duplicate file and coverage names to the dlg2sds directory. This will cause errors and result in an unsuccessful completion of the batch process.  

After the process is complete, the user will see the SDS/FMS data sets in two forms, and ARC/INFO coverage and  ARC/INFO export files (*.e00). As each DLG source file goes through the conversion process, it is queried for features that are included in SDS/FMS. Following the conversion process the process will yield SDS/FMS compliant data sets for features found in the DLG source files. Figure 8 shows the coverages and export files created for the DLG source files shown in Figure 1.

In addition to the SDS/FMS coverages and export files there will be residual coverages and export files that are a raw conversion of the DLG source files to ARC/INFO coverages. These data sets are retained for purposes of subsequent utilization. It is possible that residual coverages would be of value to the user, in that they could contain features of interest that are not documented in SDS/FMS. Residual coverages and export files will retain the same name as the DLG source file name, prefixed by the L character, designating a line coverage, or P character, designating a point or polygon coverage. Residual coverage names beginning with characters XX are temporary coverages created during the conversion can be deleted.  Residual coverages can be deleted at the user’s discretion. 

The conversion process will also create a series of text files with an *.opf suffix.  Named for coverages processed in the conversion, the files contain major-minor codes as used to assign features entities in SDS/FMS. When the conversion process is complete, the *.opf files can be deleted.

Figure 8 SDS/FMS compliant coverages and export files created by DLG to SDS
Appendix A

Below is detailed documentation of the Arc Macro Language (AML) scripts used for Converting Digital Line Graph Data into SDS/FMS compliant coverages for ESRI ARC/INFO Workstation. Twelve AML scripts are included to convert DLG data sets to SDS/FMS compliant coverages. Each AML script is documented below.

dlg2sdscon.aml

Initiates the conversion process and invokes a graphic user interface pulldown menu. Utilizes dlggui.menu to display the menu choices and correlate them with a processing AML for each Step required for the conversion procedure.

cdo2dlgren.aml
Renames all files in the workspace with a .cdo extension to a file name with a .dlg extension.

opt2dlgren.aml
Renames all files in the workspace with a .opt extension to a file name with a .dlg extension.

deletedlgs.aml

Deletes all files in the workspace with a .dlg extension.

Batch.aml

Runs six AML scripts in sequence to make the DLG to SDS/FMS data conversion.

dlgsnc.aml

dlgsepMMcodes.aml

d2spoint.aml

d2spoly.aml

d2sarc.aml

e00exporter.aml

Dlgsnc.aml 

Looks for files in the workspace with a .dlg extension and initiates a processing loop. Each source file is processed by ARC/INFO’s DLGARC command to convert the graphic features to coverage features. The AML then cleans and builds each coverage to create topology. Upon conversion to ARC/INFO coverage’s, DLG source files create acode tables (line attributes) and pcode tables (point-polygon attributes). These are joined with the feature geometry to describe the features with major-minor codes (mmpairs), expressed as integers by USGS standards. The AML executes a JOINITEM command, utilizing the cover-id field to append mmpairs to each feature in the data set.

The final step in the AML is to create and populate INFO tables with lookup relationships between the DLG mmpairs and SDS/FMS entity class and coverage names. The dbaseinfo command is issued to read the DBASE IV tables (sds1.dbf through sds5.dbf) from disk into an ARC/INFO database structure. Mmpair codes 1 through 5 are utilized because identical graphic features can represent multiple entity types of the SDS/FMS. 

DlgsepMMcodes.aml 

Relates and populates the feature codes from the SDS INFO files to the appropriate entity class and coverage names. The mmpair fields are concatenated, separated by a dash, to create single field value to join with the SDS/FMS INFO tables. For example, if a record holds major1=50 and minor1=200, the mmcode field will be equal to 50-200 (fifty dash 200). Mmcode = 50-200 is assigned a one to one relationship to entity class name “shoreline” and coverage name “hycznshr”. The final procedure of this AML joins the SDS/FMS entity class and coverage names to the coverages attribute table.

Remove_blanks.aml 

Assures that when the character field is queried for mmcode text string, the appropriate features will be selected. The processing used to create and convert the DLG data to coverages can produce spaces or trailing blanks within attribute records. This aml is used to remove extraneous spaces, blanks, or null characters from each attribute record as a safety precaution.

D2spoly.aml

Operates on polygon coverages in the workspace to identify and extract features appropriate to the SDS/FMS standards. The AML goes through each coverage to query for SDS/FMS compliant attributes and places them into a new, separate coverage. The AML reads polylist.sds to obtain the correct SDS/FMS coverage names for the features found within. For example, bdjurdsv is the first name in the polylist.sds file; the AML queries for sdspoly = “bdjurdsv”.  If a feature is found, it is extracted out to a new coverage, named “bdjurdsv”.

D2sarc.aml

Operates on identical to d2spoly.aml, but works on line coverages in the workspace. Identifies and extracts features from raw coverages appropriate to the SDS/FMS standards. The AML goes through each coverage to query for SDS/FMS compliant attributes and places them into a new, separate coverage. The AML reads linelist.sds to obtain the correct SDS/FMS coverage names for the features found within coverage. For example, bdjurdsv is the first name in the linelist.sds file; the AML queries for sdsline = “bdjurbdr”.  If a feature is found, it is extracted out to a new coverage, named “bdjurbdr”.
D2spoint.aml

Operates on identical to d2spoly.aml, but works on point coverages in the workspace. Identifies and extracts features from raw coverages appropriate to the SDS/FMS standards. The AML goes through each coverage to query for SDS/FMS compliant attributes and places them into a new, separate coverage. The AML reads pointlist.sds to obtain the correct SDS/FMS coverage names for the features found within coverage. For example, bggenexp is the first name in the pointlist.sds file; the AML queries for sdspoint = “bggenexp”.  If a feature is found, it is extracted out to a new coverage, named “bggenexp”.

E00exporter.aml
Creates an ARC/INFO interchange file (*.e00) for each coverage in the workspace.

Appendix B

Below is detailed printout of the Arc Macro Language (AML) scripts used for Converting Digital Line Graph Data into SDS/FMS compliant coverages for ESRI ARC/INFO Workstation. Twelve AML scripts are included to convert DLG data sets to SDS/FMS compliant coverages. Each complete AML script is printed below.

/*  dlg2sdscon.aml

/*  Written by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/*  Displays DLG to SDS Conversion pulldown menu

/*  This AML requires dlggui.menu

/* Sets to terminal environment variable 

/* to display the DLG conversion menu

&terminal 9999

&menu dlggui

/*  cdo2clgren.aml

/*  Written by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/* Rename all files in a directory with a *.cdo extension to *.dlg extension

&sys ren *.cdo *.dlg

&type ********* RENAME PROCESS COMPLETE *********

/*  opt2clgren.aml

/*  Written by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/* rename all files in the current directory with a *.opt extension to *.dlg extension

&sys ren *.opt *.dlg

&type ********* RENAME PROCESS COMPLETE *********

&return

/*  deletedlgs.aml

/*  Written by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/* delete all files in the current directory with a *.dlg extension

&sys DEL *.dlg

&type ********* DELETE PROCESS COMPLETE *********

&return

/*  batch.aml

/*  Written by Adam Dunlap, edited by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/*  Performs sequence of procedures to convert DLG data to SDS compliant coverages

&r dlgsnc

&r dlgsepMMcodes

&r d2spoint

&r d2spoly

&r d2sarc

&r e00exporter

&type ********* DLG TO SDS CONVERSION PROCESS COMPLETE *********

&return

/*  dlgsnc.aml

/*  Written by Adam Dunlap, edited by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/*  Run DLGARC command for every dlg file in the workspace

/*  Clean and build coverages, join attribute tables

/*  Read dbase-iv files into INFO tables 

/*  Containing major#-minor# pair codes and entity class name relationships

&echo &off

/*makes a file list from the *.dlg files in the directory

&s covers := [listfile *.dlg -file]

&if [null %covers%] &then 

&return No dlg files found... check your workspace.

/*perform DLGARC command on all dlg files in the directory

&s num := [token %covers% -count]

&do I := 1 &to %num%

&type getting Arc polygons [extract %i% %covers%]

&s go2 = [extract %i% %covers%]   

&s goo = [before %go2% .]

/* &sys chop 80 %goo%.dlg %goo%.bgr

      dlgarc optional %goo%.dlg l%goo% x%goo% attributed

      copy l%goo% p%goo%

      clean p%goo%

      build p%goo%

      build l%goo% line

      build x%goo% point

&severity &error &ignore

/*join polygon and line attribute tables

      joinitem  p%goo%.pat p%goo%.pcode p%goo%.pat p%goo%-id 

      joinitem l%goo%.aat l%goo%.acode l%goo%.aat l%goo%-id

      joinitem x%goo%.pat x%goo%.xcode x%goo%.pat x%goo%-id

        /*&s amount = [show number select]

&end                                        /* for the initial if then

/* read the sds dbase files into an info table

 dbaseinfo sds1.dbf sds1.sds

 dbaseinfo sds2.dbf sds2.sds

 dbaseinfo sds3.dbf sds3.sds

 dbaseinfo sds4.dbf sds4.sds

 dbaseinfo sds5.dbf sds5.sds

&type ********* DLG TO ARC CONVERSION PROCESS COMPLETE *********

&return

/*  dlgsepMMcodes.aml

/*  Written by Adam Dunlap, edited by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/*  Concatenate major# and minor# attribute codes from DLG features into a single field, MMCODE

/*  If multiple major#-minor# codes exist, additional fields are created , up to major5-minor5 (mmcode5)

&echo &off

/*

/*  Doing all the arc (line) coverages first

/*  Make a list of line coverages to process

&s covers := [listfile * -cover -arc]

&if [null %covers%] &then 

&return No coverages found... check your workspace.

&s num := [token %covers% -count]

&do I := 1 &to %num%

 &type getting Arc polygons [extract %i% %covers%]

 &s go4it = [extract %i% %covers%]   

 &s rv = [listitem %go4it% -arc -all -integer %go4it%.opf]

 &s type = .AAT

 &call coninfo

 &end

/*////////////////////////////////////////////////////////////////////////////

/*///////////////////////////////////////////////////////////////////////////

/*

/*   Last is the point covers

/*

&s covers := [listfile * -cover -point]

&if [null %covers%] &then 

&return No coverages found... check your workspace.

&s num := [token %covers% -count]

&do I := 1 &to %num%

 &type getting Arc polygons [extract %i% %covers%]

 &s go4it = [extract %i% %covers%]   

 &s rv = [listitem %go4it% -point -all -integer %go4it%.opf]

 &s type = .PAT 

      &call coninfo  

&end

/*

/*  Next is the polygon covers

/*

&s covers := [listfile * -cover -polygon]

&if [null %covers%] &then 

&return No coverages found... check your workspace.

&s num := [token %covers% -count]

&do I := 1 &to %num%

 &type getting Arc polygons [extract %i% %covers%]

 &s go4it = [extract %i% %covers%]   

 &s rv = [listitem %go4it% -polygon -all -integer %go4it%.opf]

 &s type = .PAT

     &call coninfo

&end    

&return

/*  ///////////////////////////////////////////////////////////////////////////////// 

/*  \\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\

&routine coninfo

&severity &error &ignore

&if %rv% = 0 &then &do

     [delete %go4it%.opf]

     kill %go4it% all

     &end

/*222222222222222222222222222222222222222222222222222

     &if %rv% = 2 &then &do

     dropitem %go4it%%type% %go4it%%type% Mmcode

     additem %go4it%%type% %go4it%%type% Mmcode 20 20 C #

     &s g4i := [upcase %go4it%]

      &data arc info 

      ARC

      SELECT %g4i%%type%

      CONCATENATE MMCODE FROM MAJOR1,'-',MINOR1

      Q STOP

      &end

     &r Remove_blanks %g4i%%type% MMCODE

     joinitem %go4it%%type% sds1.sds %go4it%%type% Mmcode 

     &end

/*4444444444444444444444444444444444444444444444444444444444

     &if %rv% = 4 &then &do

     dropitem %go4it%%type% %go4it%%type% Mmcode

     dropitem %go4it%%type% %go4it%%type% Mmcode2

     additem %go4it%%type% %go4it%%type% Mmcode 20 20 C # 

     additem %go4it%%type% %go4it%%type% Mmcode2 20 20 C #  

      &s g4i := [upcase %go4it%]

      &data arc info

      ARC

      SELECT %g4i%%type%

      CONCATENATE MMCODE FROM MAJOR1,'-',MINOR1

      CONCATENATE MMCODE2 FROM MAJOR2,'-',MINOR2

      Q STOP

      &end

     &r Remove_blanks %g4i%%type% MMCODE

     &r Remove_blanks %g4i%%type% MMCODE2

     joinitem %go4it%%type% sds1.sds %go4it%%type% Mmcode 

     joinitem %go4it%%type% sds2.sds %go4it%%type% Mmcode2

     &end

/*666666666666666666666666666666666666666666666666666666666666

     &if %rv% = 6 &then &do

     dropitem %go4it%%type% %go4it%%type% Mmcode

     dropitem %go4it%%type% %go4it%%type% Mmcode2

     dropitem %go4it%%type% %go4it%%type% Mmcode3

     additem %go4it%%type% %go4it%%type% Mmcode 20 20 C # 

     additem %go4it%%type% %go4it%%type% Mmcode2 20 20 C # 

     additem %go4it%%type% %go4it%%type% Mmcode3 20 20 C # 

     &s g4i := [upcase %go4it%]

      &data arc info

      ARC

      SELECT %g4i%%type%

      CONCATENATE MMCODE FROM MAJOR1,'-',MINOR1

      CONCATENATE MMCODE2 FROM MAJOR2,'-',MINOR2

      CONCATENATE MMCODE3 FROM MAJOR3,'-',MINOR3

      Q STOP

      &end

       &r remove_blanks %g4i%%type% MMCODE

       &r remove_blanks %g4i%%type% MMCODE2

       &r remove_blanks %g4i%%type% MMCODE3

          joinitem %go4it%%type% sds1.sds %go4it%%type% Mmcode 

          joinitem %go4it%%type% sds2.sds %go4it%%type% Mmcode2

          joinitem %go4it%%type% sds3.sds %go4it%%type% Mmcode3

     &end

/***8888888888888888888888888888888888888888888888888888888888

     &if %rv% = 8 &then &do

     dropitem %go4it%%type% %go4it%%type% Mmcode

     dropitem %go4it%%type% %go4it%%type% Mmcode2

     dropitem %go4it%%type% %go4it%%type% Mmcode3

     dropitem %go4it%%type% %go4it%%type% Mmcode4

     additem %go4it%%type% %go4it%%type% Mmcode 20 20 C # 

     additem %go4it%%type% %go4it%%type% Mmcode2 20 20 C # 

     additem %go4it%%type% %go4it%%type% Mmcode3 20 20 C # 

     additem %go4it%%type% %go4it%%type% Mmcode4 20 20 C #

     &s g4i := [upcase %go4it%]     

      &data arc info

      ARC

      SELECT %g4i%%type%

      CONCATENATE MMCODE FROM MAJOR1,'-',MINOR1

      CONCATENATE MMCODE2 FROM MAJOR2,'-',MINOR2

      CONCATENATE MMCODE3 FROM MAJOR3,'-',MINOR3

      CONCATENATE MMCODE4 FROM MAJOR4,'-',MINOR4

      Q STOP

      &end

       &r remove_blanks %g4i%%type% MMCODE

       &r remove_blanks %g4i%%type% MMCODE2

       &r remove_blanks %g4i%%type% MMCODE3

       &r remove_blanks %g4i%%type% MMCODE4     

          joinitem %go4it%%type% sds1.sds %go4it%%type% Mmcode 

          joinitem %go4it%%type% sds2.sds %go4it%%type% Mmcode2

          joinitem %go4it%%type% sds3.sds %go4it%%type% Mmcode3

          joinitem %go4it%%type% sds4.sds %go4it%%type% Mmcode4

     &end

     &return

     &if %rv% = 10 &then &do

     dropitem %go4it%%type% %go4it%%type% Mmcode

     dropitem %go4it%%type% %go4it%%type% Mmcode2

     dropitem %go4it%%type% %go4it%%type% Mmcode3

     dropitem %go4it%%type% %go4it%%type% Mmcode4

     dropitem %go4it%%type% %go4it%%type% Mmcode5

     additem %go4it%%type% %go4it%%type% Mmcode 20 20 C # 

     additem %go4it%%type% %go4it%%type% Mmcode2 20 20 C # 

     additem %go4it%%type% %go4it%%type% Mmcode3 20 20 C # 

     additem %go4it%%type% %go4it%%type% Mmcode4 20 20 C # 

     additem %go4it%%type% %go4it%%type% Mmcode5 20 20 C # 

     &s g4i := [upcase %go4it%]

     &data arc info

     ARC

     SELECT %g4i%%type%

     CONCATENATE MMCODE FROM MAJOR1,'-',MINOR1

     CONCATENATE MMCODE2 FROM MAJOR2,'-',MINOR2

     CONCATENATE MMCODE3 FROM MAJOR3,'-',MINOR3

     CONCATENATE MMCODE4 FROM MAJOR4,'-',MINOR4

     CONCATENATE MMCODE5 FROM MAJOR5,'-',MINOR5

     Q STOP

     &end

       &r remove_blanks %g4i%%type% MMCODE

       &r remove_blanks %g4i%%type% MMCODE2

       &r remove_blanks %g4i%%type% MMCODE3

       &r remove_blanks %g4i%%type% MMCODE4     

       &r remove_blanks %g4i%%type% MMCODE5    

          joinitem %go4it%%type% sds1.sds %go4it%%type% Mmcode 

          joinitem %go4it%%type% sds2.sds %go4it%%type% Mmcode2

          joinitem %go4it%%type% sds3.sds %go4it%%type% Mmcode3

          joinitem %go4it%%type% sds4.sds %go4it%%type% Mmcode4

          joinitem %go4it%%type% sds5.sds %go4it%%type% Mmcode5

     &end

&return

/*************************************************************************************

/************************************************************************************

&type ********* Major-Minor Code Attribution Complete *********

&return

/*  remove_blanks.aml

/*  Written by Environmental Systems Research Institute, Inc.  

/*  Copyright 1995

/*  This aml uses an INFO program to remove blanks from an item

&severity &error &routine bailout 

&args info_file item 

/* Check for missing arguments

&if [null %info_file%] or [null %item%] &then

  &return Usage: &R REMOVE_BLANKS <info file> <item to remove spaces from> 

/* Check if info file and item exist, and if the item is character.

&if ^ [exists %info_file% -info] &then 

  &return INFO file %info_file% does not exist

&if ^ [iteminfo %info_file% -info %item% -exists] &then 

    &return %item% does not exist within INFO file %info_file%

&else

  &if [iteminfo %info_file% -info %item% -redefined] eq .true. &then

    &return This program will not work on redefined items

  &else

    &if [extract 3 [iteminfo %info_file% -info %item% -definition]] ne 'C' &then

      &return Item %item% is not a CHARACTER item

/* Find the item width

&s itm_width = [extract 1 [iteminfo %info_file% -info %item% -definition]]

/* Check if the width is 1.  

&if %itm_width% <= 1 &then

  &return Item width must be 2 characters or larger.

&type Removing spaces from %item%...

&data arc info

arc

&if [exists rmvspcs -info] &then &s del$stat [delete rmvspcs -info]

REM

FORMAT $NUM1,1,1,I

CALC $NUM1 = 1

REM

DEFINE TEMP     

ITEM,[calc %itm_width% * 2],[calc %itm_width% * 2],C

[unquote '']

REM

ADD

A

[unquote '']

REM 

REDEFINE

1,IN,%itm_width%,%itm_width%,C

[calc %itm_width% + 1],OUT,%itm_width%,%itm_width%,C

1,IF1,1,1,C

[calc %itm_width% + 1],IF2,1,1,C

[unquote ' ']

ALTER IF1

,,,,,,,,1,%itm_width%

ALTER IF2

,,,,,,,,2,%itm_width%

REM

SEL [translate %info_file%]

REL TEMP 1 BY $NUM1 LINK

REM

PROGRAM RMVSPCS

PROGRAM SECTION EVEN

MOVE ' ' TO $1IN

MOVE [translate %item%] TO $1IN

MOVE ' ' TO $1OUT

CALC $IDX1 = 1

CALC $IDX2 = 1

DO WHILE $IDX1 LE %itm_width% 

  IF $1IF1 NE ' '

    MOVE $1IF1 TO $1IF2

    CALC $IDX2 = $IDX2 + 1

    CALC $IDX1 = $IDX1 + 1

  ELSE

    CALC $IDX1 = $IDX1 + 1

  ENDIF

DOEND

MOVE $1OUT TO [translate %item%]

PROGRAM SECTION ODD  

END

[unquote '']

COMPILE RMVSPCS

RUN RMVSPCS 

Q STOP

&end

/*

&call exit

&return /* To calling aml

/*

/* ---------------------------- Routine Exit ---------------------------------

/*

&routine exit

&if [exists temp -info] &then

  &type [delete temp -info]

&return

/*

/* --------------------------- Routine Bailout ------------------------------- 

/* &routine bailout

&severity &error &ignore

&call exit

&return; &return &error Bailing out of remove_blanks.aml

/*  d2sarc.aml

/*  Written by Adam Dunlap, edited by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/*  Select polygon features from dlg source coverages and put them into sds coverages using ArcEdit

&echo &off

&s pcovers := [listfile p*.opf -file]   /* Get the poly coverages

&if [null %pcovers%] &then  /* the first If then 

&return Error....check your workspace.

&s num := [token %pcovers% -count]

&do I := 1 &to %num%

 &type getting Poly lines [extract %i% %pcovers%]

 &s lname := [before [extract %i% %pcovers%] .]   

&severity &error &ignore

 ae

/* first you have to open the file

&s fileos := [open polylist.sds openstatus -read]

  &if %openstatus% <> 0 &then

    &return &warning Error opening file

&s ent := [read %fileos% readstatus]

  &if %readstatus% <> 0 &then

    &return &warning Error...couldn't read file

  &do &while %readstatus% = 0

 &ty .....should  be %ent%?

   &s uent := [quote %ent%]

  ec %lname%

  ef poly

  sel all

  sel Sdspoly = %uent%

  put %ent%

  y

  save

  ec %ent% poly

  build

  save

  &s ent := [read %fileos% readstatus]

  &end

 quit

/* Close the file

&if [close %fileos%] <> 0 &then

 &ty can't close file linelist

&end /*stops if then 1st

&type ********* SDS POLYGON CONVERSION PROCESS COMPLETE *********

&return

/*****************************************************************************************

/******************************************************************************************

/*  d2sarc.aml

/*  Written by Adam Dunlap, edited by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/*  Select arc features from dlg source coverages and put them into sds coverages using ArcEdit

&echo &off

&s lcovers := [listfile l*.opf -file]   /* Get the line coverages

&if [null %lcovers%] &then  /* the first If then 

&return No covers exist....check your workspace.

&s num := [token %lcovers% -count]

&do I := 1 &to %num%

 &type getting Arc lines [extract %i% %lcovers%]

 &s lname := [before [extract %i% %lcovers%] .]   

&severity &error &ignore

 ae

/* first you have to open the file

&s fileos := [open linelist.sds openstatus -read]

  &if %openstatus% <> 0 &then

    &return &warning Error opening file

&s ent := [read %fileos% readstatus]

  &if %readstatus% <> 0 &then

    &return &warning Error...couldn't read file

  &do &while %readstatus% = 0

 &ty .....should be %ent%?

   &s uent := [quote %ent%]

  ec %lname%

  ef arc

  sel all

  sel Sdsline = %uent%

  put %ent%

  y

  save

  &s ent := [read %fileos% readstatus]

  &end

 quit

/* Close the file

&if [close %fileos%] <> 0 &then

 &ty can't close file linelist

&end /*stops if then 1st

&return

/*  d2spoint.aml

/*  Written by Adam Dunlap, edited by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/*  Select point features from dlg source coverages and put them into sds coverages using ArcEdit

&s xcovers := [listfile x*.opf -file]   /* Get the point coverages

&if [null %xcovers%] &then  /* the first If then 

&return Error....check your workspace.

&s num := [token %xcovers% -count]

&do I := 1 &to %num%

 &type getting Arc points called labels [extract %i% %xcovers%]

 &s lname := [before [extract %i% %xcovers%] .]   

&severity &error &ignore

 ae

/* first you have to open the file

&s fileos = [open pointlist.sds openstatus -read]

  &if %openstatus% <> 0 &then

    &return &warning Error opening file

&s ent := [read %fileos% readstatus]

  &if %readstatus% <> 0 &then

    &return &warning Error...couldn't read file

  &do &while %readstatus% = 0

 &ty .....should this be %ent%?

   &s uent := [quote %ent%]

  ec %lname%

  ef label

  sel all

  sel Sdspoint = %uent%

  put %ent%

  y

  save

  &s ent := [read %fileos% readstatus]

  &end

 quit

/* Close the file

&if [close %fileos%] <> 0 &then

 &ty can't close file linelist

&end /*stops if then 1st

&type ********* SDS POINT CONVERSION PROCESS COMPLETE *********

&return

/*  E00exporter.aml

/*  Written by Adam Dunlap, edited by Eric Sheehan, Michael Baker Jr., Inc.

/*  1/8/01

/*  Exports all covers in a workspace to .e00 files  

&echo &off

/*make a list of all coverages in the workspace

&s covers := [listfile * -cover -all]

&if [null %covers%] &then 

&return No coverages here....check your workspace.

/*For each cover in the workspace make an export file.

&s num := [token %covers% -count]

&do I := 1 &to %num%

&s go2 = [extract %i% %covers%]  

export cover %go2% %go2%

&end

&type ********* EXPORT PROCESS COMPLETE *********

&return

Figure 9 Arc Macro Language (AML) Scripts Process Flowchart

Appendix D

ESRI Spatial Data Transfer Standards Document Reference List

ArcDoc Online User Help  – ESRI ARC/INFO Software Users Manual, copyright 2000 Environmental Research Institute Inc. 

ESRI Spatial Data Transfer Standards Software Pricing

Software
Price
Notes

ARC/INFO
$14,000


Appendix E
ESRI DLG2SDS Directory Structure

AML 


Contains all AML scripts and supporting conversion files for the DLG to SDS/FMS Conversion.



batch.aml

cdo2dlgren.aml




d2sarc.aml

d2spoint.aml




d2spoly.aml

d2sds.aml




dlg2sdscon.aml

dlgsepmmcodes.aml




dlgsnc.aml

e00exporter.aml




opt2dlgren.aml

remove_blanks.aml

dlggui.menu

polylist.sds 

linelist.sds 

pointlist.sds

sds1.dbf


sds2.dbf

sds3.dbf


sds4.dbf

sds5.dbf 
DLGSOURCE


Contains sample DLG source input files for New Mexico as presented in the User Guide document.


504043.hy.opt  

hydrography


504042.hy.opt

hydrography


504051.rr.opt

rail road


504054.rr.opt

rail road


504059.mt.opt

pipelines and transit lines


504060.mt.opt

pipelines and transit lines


504066.rd.opt

roads and trails


504067.rd.opt

roads and trails


504076.pl.opt

public land surveys


504077.pl.opt

public land surveys


504083.bd.opt

boundaries


504085.bd.opt

boundaries
DOCS


Contains User Guide Document.


EDIT FOR INDIVIDUAL DOCUMENT NAMES

OUTPUT


ARC/INFO workspace containing GIS data sets converted using the *.dlg source files in the DLGSOURCE directory.

README.TXT


This readme file.

